Проверка арифметического выражения на корректность

<выражение>::=<терм>|<терм>+<выражение>|<терм>-<выражение>

<терм>::=<множитель>|<множитель>*<терм>|<множитель>/<терм>

<множитель>::=(<выражение>)|<имя>|<натуральное число>

<имя>::=<буква>|<имя><буква>|<имя><цифра>

<натуральное число>::=<цифра>|<натуральное число><цифра>

<цифра>::=0|1|2|3|4|5|6|7|8|9

<буква>::=_|A|B|…|Z|a|b|…|z|

Здесь слева от метазнака ::= стоит определяемое понятие, а справа — его определение. Знаки | обозначают логическую операцию ИЛИ в определении, остальные символы входят в определение того или иного понятия. В так введенной грамматике для арифметического выражения отсутствуют унарные операции.

	Конечным автоматом называется реализация алгебраической структуры 
(S, (, m, s0, F), где

· S — непустое множество состояний;
· ( — конечное множество входных символов (алфавит);
· m — отображение S(((S, или функция переходов, которая каждой паре (символ, состояние) ставит в соответствие состояние из множества S;
· s0 — состояние из S, известное как начальное (стартовое);
· F — множество заключительных (допускающих) состояний, F(S или просто соответствует окончанию просмотра текста.
Иногда сюда добавляют для каждой пары (символ, состояние) процедуру обработки символа (например, его печать). Работа автомата заключается в том, что изначально автомат находится в состоянии s0 и под действием первого входного символа переходит в следующее состояние и читает следующий символ и т.д. Автомат заканчивает свою работу, если достигнуто одно из состояний множества F, или прочитан символ, не принадлежащий (, или входные данные исчерпаны.

Если отображение m однозначно, то есть каждой паре (символ, состояние) соответствует определенное состояние, то автомат называют детерминированным, в противном случае (одной и той же паре в соответствие ставится сразу несколько состояний, чаще всего в зависимости от предыдущих или последующих символов обрабатываемой входной строки) — недетерминированным.

Автомат проверки корректности

Сост.           Символ       Действие    Новое состояние
0    нет символов         2

0       (        k:=k+1   0

0    буква, цифра         1

0    другие символы       ошибка

1    нет символов         2

1        )      k:=k-1    1

1    знак операции        0

1    другие символы       ошибка

2    буква, цифра, )   корректно(k=0)

2    другие символы       ошибка
	const digits : set of char=['0'..'9'];

letters:set of char=['_','A'..'Z','a'..'z'];

     op : set of char=['+','-','*','/'];

var s : string;

  i,k : integer;

  state : 0..3;

procedure error;

begin
  writeln('Выражение некорректно'); halt

end;

procedure Identifier; {пропустить имя}

begin
  while (i<length(s)) and 

      (s[i+1] in (letters+digits)) do i:=i+1

end;

procedure Number; {пропустить число}

begin
   while (i<length(s)) and 

         (s[i+1] in digits) do i:=i+1

end;

begin {Main}

  readln(s);

  i:=0; k:=0; state:=0;

  while state<>3 do
  case state of
  0:if i<length(s) then
    begin
      i:=i+1;

      if s[i]='(' then k:=k+1 else
      if not (s[i] in (letters+digits))

         then error else
      begin
        if s[i] in letters then Identifier

        else if s[i] in digits then Number;

        state:=1

      end
    end else state:=2;

  1:if i<length(s) then
    begin
      i:=i+1;

      if s[i]=')' then
        begin
          k:=k-1;

          if k<0 then error

        end
      else
      if s[i] in op then state:=0 else error

    end else state:=2;

  2:if (s[i] in(letters+digits+[')']))

       and(k=0) then
     begin
       writeln('Выражение корректно');

       state:=3

     end else error

  end {case}

end.

	"Палочный" способ разбора арифметических выражений

Сначала ищется операция, которую можно выполнить первой, она выполняется и в измененном выражении вновь ищется первая выполнимая операция. Поставим в соответствие исходному арифметическому выражению строку, в которой каждый элементарный операнд из выражения (в случае подсчета такими операндами являются только числа) заменим на символ ‘|’ (“палочку”), а знаки операций и скобки оставим неизменными. Теперь выпишем в терминах “палочек” действия, которые следует выполнять при подсчете значения арифметического выражения.

1) (|) ( |

2) |*| или |/| ( |
3) (|(|) ( (|)или (|(|( ( (|(
4) |(| ( |
Применять указанные правила следует так. В “палочной”строке ищется первое вхождение подстроки сначала из левой части правила 1, если такая не найдется вообще — то из правила 2 и т.д. Для найденной подстроки осуществляется замена согласно правилу, а над соответствующими этой подстроке элементами исходного выражения производятся те же арифметические действия, что и в найденной подстроке (для правила 1 — просто снимаются скобки). После этого то же самое правило пытаются применить еще раз, а если это невозможно, то снова переходят к поиску подстроки из правила 1 (а не из следующего правила, что весьма существенно). Можно доказать, что таким образом к поиску подстроки из левой части правила 4 мы приступим лишь тогда, когда в выражении уже не останется ни скобок, ни операций умножения или деления. Действия заканчиваются, когда у нас останется одна палочка, т.е. исходное выражение будет сведено к одному числу, являющемуся его значением.

	Метод рекурсивного спуска

var s:string;{исходное выражение}

i:integer;{номер текщего симвовла}

function Mul:longint; forward;

function Factor:longint; forward;

function Add:longint;
{суммирует слагаемые}

var q,res:longint; c:char;

begin
 res:=Mul;{первое слагаемое}

 while s[i] in ['+','-'] do
 begin
  c:=s[i]; i:=i+1;

  q:=Mul;{очередное слагаемое}

  case c of
  '+':res:=res+q;

  '-':res:=res-q;

  end
 end;{while}
 Add:=res

end;

function Mul:longint;

{перемножает множители}

var q,res:longint;

    c:char;

begin
 res:=Factor;{первый множитель}

 while s[i] in ['*','/'] do
 begin
  c:=s[i]; i:=i+1;

  q:=Factor;{очередной множитель}

  case c of
  '*':res:=res*q;

  '/':if q=0 then
      begin 

       writeln('деление на 0'); halt 

      end
      else res:=res div q

  end {case}

 end; {while}

 Mul:=res

end;

function Number:longint;

{выделяет число}

var res:longint;

begin
 res:=0;

 while s[i] in ['0'..'9'] do
 begin
 res:=res*10+(ord(s[i])-ord('0'));

 i:=i+1

 end;

 Number:=res

end;

function Factor:longint;

{выделяет множитель}

var q:longint;

    c:char;

begin
 case s[i] of
 '0'..'9':Factor:=Number;

 '(':begin i:=i+1;Factor:=Add;
     if s[i] <> ')' then
       begin writeln('ошибка'); halt end;
     i:=i+1;{пропустили ')'}end;

 '-':begin i:=i+1; Factor:=-Factor; end
  else begin writeln('ошибка'); halt end
  end {case}

end;

begin {основная программа}

 readln(s); s:=s+'.'; i:=1; r:=Add;
 if i<>length(s) then writeln('ошибка')

                 else writeln(r)
end.


